(только для членов Жюри)

Идеи решения задач II-го этапа Всероссийской олимпиады школьников по информатике

(10–11 класс, 2000–2001 учебный год, Пермская область)

1) Сосчитай. (Составитель — Гладков В.П., оппонент — Еремин Е.А.). 10 баллов.


Нужно написать программу для решения квадратного уравнения: x2-Cx+D=0.
2) Цифровой корень.  (Автор — Брызгалов Е.В., оппонент — Еремин Е.А.) 20 баллов.
Вариант решения на языке Turbo Pascal:

Program CR;

var S:string;

    i:integer;

    r:integer;

Function Sum(X:integer):integer;

begin

 if X<10 then Sum:=X

         else Sum:=X mod 10 + Sum(X div 10);

end;

begin

 assign(input,'input.txt');reset(input);

 assign(output,'output.txt');rewrite(output);

 readln(S);

 for i:=1 to length(S) do

  r:=r+ord(S[i])-48;

 while r>9 do

  r:=Sum(r);

 writeln(r);

 close(input);

 close(output);

end.

3) Вращение слова. (Составитель — Гладков В.П., оппонент — Деменев А.Г.). 35 баллов.


Задача на длинную арифметику. Нужно найти остаток от деления N на длину заданного слова. Затем выполняется циклический сдвиг на полученное число. Например, N=17, и слово Program. К – количество сдвигов = 17 mod 7=3. Выводим конец строки после первых 3 символов, затем первые три 3 символа. gramPro.

4) Маршрут кузнечика. (Составитель — Брызгалов Е.В., оппонент — Гладков В.П.). 35 баллов.
Вариант решения на языке Turbo Pascal:

Program CrassHopper;

var a:array [0..30] of longint;

    Jump:array [1..10] of integer;

    i,N,M:integer;

Function F(x:integer):longint;

var i,r:longint;

begin

 if a[x]=-1 then

  begin

   r:=0;

   for i:=1 to M do

    if x-Jump[i]>=0 then 

     r:=r+F(x-Jump[i]);

   a[x]:=r;

  end; 

 F:=a[x];

end;

begin

 assign(input,'input.txt');reset(input);

 assign(output,'output.txt');rewrite(output);

 readln(N,M);

 for i:=1 to M do

  Read(Jump[i]);

 for i:=1 to N do A[i]:=-1;

 a[0]:=1;

 writeln(F(N));

 close(input);close(output);

end.

5) Восстановление последовательности. (Составитель — Гладков В.П., оппонент — Брызгалов Е.В.). 40 баллов.

{ Решение задачи о последовательностях}

{ Проверка корректности не производится!}

const max=2000;

var a,b,c:array[1..max] of integer;

N,i,j,k,l:integer;

fin, fout:text;

begin

assign(fin,'input.txt');

reset(fin);

N:=0;

while not(eof(fin)) do

 begin

  inc(N);

  read(fin,a[N]);

  if eoln(fin) then readln(fin);

 end;

close(fin);

for i := 1 to N do b[i]:=N-i; {Сколько есть чисел меньше данного}

for i:=N downto 1 do {ищем начиная с конца}

 begin

  j:=N;

  while b[j]<a[i] do dec(j); {находим нужное число}

  c[i]:=j;

  for k:=1 to j do dec(b[k]) {Обновляем сколько осталось чисел меньше данного}

 end;

assign(fout,'output.txt');

rewrite(fout);

for i:=1 to N-1 do write(fout,c[i],' '); writeln(fout,c[N]);

close(fout);

end.
6) Инверсии в ДНК.  (Составитель — Брызгалов Е.В., оппонент — Каганов И.В.). 60 баллов.
Вариант решения на языке Turbo Pascal:

Program DNA;

Type Massiv=array [1..65535] of byte;

     TDNA=string[8];

var S,S1,S2:TDNA;

    Step:^Massiv;

    Len:integer;

    Start,Finish:word;

    Cands:array [1..5000] of word;

    iR,iW:integer;

    Sol:boolean;

    i,j:integer;

Function Counts(S:TDNA):integer;

var i,R:integer;

begin

 i:=0;

 r:=0;

 For i:=1 to len do

  case S[i] of

  'А':r:=r+1000;

  'Г':r:=r+100;

  'Ц':r:=r+10;

  'Т':r:=r+1

  end;

 Counts:=r;

end;

Function DNAToInt(S:TDNA):word;

var r:word;i:integer;

begin

 r:=0;

 for i:=1 to len do

  begin

   r:=r*4;

   case S[i] of

  'А':;

  'Г':r:=r+1;

  'Ц':r:=r+2;

  'Т':r:=r+3

   end;

  end;

 DNAToInt:=r;

end;

Function IntToDNA(X:word):TDNA;

var r:TDNA;i:integer;

begin

 r:='';

 for i:=1 to len do

  begin

   case (X mod 4) of

   0:r:='А'+r;

   1:r:='Г'+r;

   2:r:='Ц'+r;

   3:r:='Т'+r;

   end;

  X:=X div 4;

  end;

 IntToDNA:=r;

end;

Function Inverse(X:TDNA;i,j:integer):TDNA;

var r:TDNA;k:integer;

begin

 r:=X;

 for k:=i to j do

  r[k]:=X[i+j-k];

 Inverse:=r;

end;

Procedure Check(X:TDNA);

var C:word;

begin

 C:=DNAToInt(X);

 if step^[C]=255 then

   begin

    step^[C]:=step^[Cands[iR]]+1;

    Cands[iW]:=C;

    inc(iW);

    if C=Finish then Sol:=True;

   end;

end;

begin

 assign(input,'input.txt');reset(input);

 assign(output,'output.txt');rewrite(output);

 Readln(S1);

 if S1='АААААААА' then S1:='ТТТТТТТТ';

 Readln(S2);

 if S2='АААААААА' then S2:='ТТТТТТТТ';

 Len:=Length(S1);

 if Counts(S1)<>Counts(S2)

 then writeln(-1)

 else

  begin

   Start:=DNAToInt(S1);

   Finish:=DNAToInt(S2);

   New(step);

   FillChar(step^,sizeof(step^),255);

   step^[Start]:=0;

   Cands[1]:=Start;

   iR:=1;iW:=2;Sol:=False;

   if Start=Finish then Sol:=True;

   repeat

    S:=IntToDNA(Cands[iR]);

    for i:=1 to Len-1 do

     for j:=i+1 to Len do

      begin

       Check(Inverse(S,i,j));

      end;

    inc(iR);

   until Sol;

   writeln(step^[Finish]);

   Dispose(step);

  end;

  close(input);

  close(output)

end.

